

Monte Carlo Simulation of the 2D Ising Model

The Metropolis Algorithm

We know that the expectation value of an observable A can be written as

$$\langle A \rangle = \frac{\sum_r A_r e^{-\beta E_r}}{\sum_r e^{-\beta E_r}}, \quad (1)$$

where A_r is the value of A for the state r . So given a system that has a discrete number of states, we could, using a computer, calculate A for each state and weight these values by their Boltzmann factors to find the average A . This might be feasible for a system with a small number of states, but if we have a 20×20 spin lattice interacting via the Ising model, there are 2^{400} states, so we cannot possibly examine all of them.

What if we decide to just sample some of the states? How would we decide which ones? This is where the “Monte Carlo” part comes in. Named for the Mediterranean casino town, a Monte Carlo method is any algorithm that involves a pseudorandom number generator.

One (bad) way of using random numbers would be to randomly pick a lot of states, measure A for each of them, and weight these values of A by their Boltzmann factors. We might get close to the right answer if we sampled a lot of states, but we would spend a lot of time calculating A for states that contribute very little to the final result (an Ising lattice at very high temperature is unlikely to be in the state with all spins pointing in one direction).

Instead of sampling (measuring parameters like A for) a lot of states and then weighting them by their Boltzmann factors, it makes more sense to *choose states based on their Boltzmann factors* and then weight them equally. This is known as the Metropolis algorithm, which is an *importance sampling* technique. One pass through the algorithm is described here:

1. A trial configuration is made by randomly choosing one spin.
2. The energy difference of the trial state relative to the present state, δE , is calculated.
3. If $\delta E \leq 0$, the trial state is energetically favorable and thus accepted. Otherwise, a random number $0 \leq \eta \leq 1$ is generated, and the new state is only accepted if $\exp(-\beta \delta E) > \eta$. This condition can be rewritten as $-\beta \delta E > \log \eta$, which is what I used in the code.

Calculating Observables

We can obtain some qualitative information about our simulation by watching the spin array during a simulation. I have written an IDL program, `see_spins.pro`, that allows us to do this. For high temperatures, the spins remain randomly aligned after long periods of equilibration, whereas for low temperatures, the spins end up pointing in mostly the same direction.

To get more quantitative results, we can measure the energy and the magnetization at each step of the routine. Before we start taking statistics, we should allow the system to equilibrate for a long time (my code equilibrates for `nequil` passes). We can then measure the magnetization by taking the sum of all the spins in the lattice, and we can calculate the energy by determining the energy for each spin and dividing by two for double counting.

What about the specific heat or susceptibility? There isn't a good way to calculate a derivative of the partition function in our code, but it turns out that the specific heat can also be written in terms of the variance of the energy:

$$\begin{aligned}
 C_V &= \frac{\partial \langle E \rangle}{\partial T} \\
 &= -\frac{\beta}{T} \frac{\partial \langle E \rangle}{\partial \beta} \\
 &= \frac{\beta}{T} \frac{\partial^2 \ln Z}{\partial \beta^2} \\
 &= \frac{\beta}{T} \frac{\partial}{\partial \beta} \left(\frac{1}{Z} \frac{\partial Z}{\partial \beta} \right) \\
 &= \frac{\beta}{T} \left[\frac{1}{Z} \frac{\partial^2 Z}{\partial \beta^2} - \frac{1}{Z^2} \left(\frac{\partial Z}{\partial \beta} \right)^2 \right] \\
 &= \frac{\beta}{T} \left[\langle E^2 \rangle - \langle E \rangle^2 \right].
 \end{aligned} \tag{2}$$

Incidentally, this is known as the Fluctuation Dissipation Theorem.

Similarly, the magnetic susceptibility, χ , can be written in terms of the variance in the magnetization:

$$\begin{aligned}
 \chi &= \frac{\partial \langle M \rangle}{\partial H} \\
 &= \beta \left[\langle M^2 \rangle - \langle M \rangle^2 \right].
 \end{aligned} \tag{3}$$

So by keeping statistics on E , E^2 , M , and M^2 , we can plot the energy, the magnetization, the specific heat, and the magnetic susceptibility. On each of these graphs, each circle represents an independent run of 100,000 steps of equilibration and 100,000 more steps of data taking.




Figure 1: The energy is a continuous function of temperature, which, as we expect, increases as a function of T .




Figure 2: The magnetization drops off sharply near the critical temperature, which, in our units where $k = J = 1$, is approximately 2.3.




Figure 3: The specific heat has a peak at the critical temperature.




Figure 4: The magnetic susceptibility has a sharp jump at the critical temperature.

Codes

This FORTRAN 90 code generates statistics on energy, heat capacity, magnetization, and magnetic susceptibility for a range of temperatures:

```
1  program ising ! 2D Monte Carlo Simulation of Ising Model
2
3  ! Lisa Larrimore, lisal@sccc.swarthmore.edu
4  ! 3 May 2002
5  ! Physics 114 Final Project
6
7  ! This program is adapted from the Ising Model program written in
8  ! BASIC by Elaine Chandler that appears on p. 184 of David Chandler's
9  ! Introduction to Modern Statistical Mechanics.
10
11 ! The input parameters for this program are in "ising.in", and they
12 ! allow the size, length, and initial configuration of the simulation
13 ! to be changed. See comments in file.
14
15 ! This program has three output files:
16 !
17 !   "spin-array"           Contains snapshots of the spin lattice at the end of
18 !                           each temperature run (or throughout the middle of the
19 !                           run, if only looking at one temperature). Can be
20 !                           visualized with the IDL program see_spins.pro
21 !
22 !   "magnetization"       Contains four columns: each temperature, the
23 !                           average magnetization at that temp, the ave magnetizaion
24 !                           squared at that temp, and the susceptibility.
25 !
26 !   "energy"              Contains four columns: each temperature, the
27 !                           average energy at that temp, the ave energy squared
28 !                           at that temp, and the heat capacity.
29
30 implicit none
31
32 ! Variable declarations:
33 integer :: i, j, m, n, m2, n2 ! dummy integers
34 integer, allocatable :: A(:, :) ! matrix containing spins
35 integer :: nrows, ncols ! number of rows and cols of A
36 real :: temp, beta ! temperature, inverse temperature
37 integer :: ConfigType ! starting configuration type
38 integer :: npass ! number of passes for MC algorithm
39 integer :: ipass ! the current pass number
40 integer :: nequil ! number of equilibration steps
41 integer :: trial_spin ! values of changed spin
42 real :: high_temp ! starting temp for scan
43 real :: low_temp ! final temp for scan
44 real :: temp_interval ! interval between scan points
45 integer :: nscans ! number of scans (each at diff T)
46 integer :: iscan ! current scan number
47 logical :: MovieOn ! set to .true. to make movie of 1 temp
48 real :: deltaU ! change in energy between 2 configs
```

```

49  real :: deltaU1, deltaU      ! energy changes for lattice gas
50  real :: log_eta             ! log of random number to compare to
51  real :: magnetization       ! magnetization of all spins in lattice
52  real :: magnetization_ave   ! cumulative average magnetization
53  real :: magnetization2_ave  ! cumulative average of mag. squared
54  real :: energy              ! energy of all spins in lattice
55  real :: energy_ave          ! cumulative average of energy
56  real :: energy2_ave         ! cumulative average of energy squared
57  integer :: output_count     ! # times things have been added to averages
58
59  print*, "_____MONTE CARLO 2D ISING MODEL_____"
60  print*, "Monte Carlo Statistics for 2D Ising Model with"
61  print*, "  periodic boundary conditions."
62  print*, "The critical temperature is approximately 2.3, as seen on"
63  print*, "  Chandler p. 123."
64
65  ! Read in input parameters from file "ising.in"
66  open(unit=11,file='ising.in',status='old',action='read')
67  read(11,*) ; read(11,*) nrows
68  read(11,*) ; read(11,*) ncols
69  read(11,*) ; read(11,*) npass
70  read(11,*) ; read(11,*) nequil
71  read(11,*) ; read(11,*) high_temp
72  read(11,*) ; read(11,*) low_temp
73  read(11,*) ; read(11,*) temp_interval
74  read(11,*) ; read(11,*) ConfigType
75  read(11,*) ; read(11,*) MovieOn
76  close(11)
77
78  ! Set the dimensions of the matrix of spin arrays. This program uses
79  ! periodic boundary conditions, so the first two rows and columns are
80  ! the same as the last two.
81  allocate(A(nrows+2,ncols+2))
82
83  ! Open output files:
84  open(unit=32,file='spin-array',status='replace',action='write')
85  write(32,*) nrows
86  write(32,*) ncols
87  nscans = int((high_temp - low_temp)/temp_interval) + 1
88  if (MovieOn) then
89    write(32,*) 51
90    write(32,*) 1
91  else
92    write(32,*) nscans
93    write(32,*) 2
94  endif
95
96  open(unit=33,file='magnetization',status='replace',action='write')
97  write(33,*) "temp  ave_magnetization  ave_magnetization^2  susceptibility"
98  open(unit=34,file='energy',status='replace',action='write')
99  write(34,*) "temp  ave_energy  ave_energy^2  C_v"
100
101  scan_loop: do iscan = 1, nscans

```

```

102     temp = high_temp - temp_interval*(iscan-1)
103     print*, "Running program for T =", temp
104
105     ! Initialize variables
106     beta = 1.0/temp
107     output_count = 0
108     energy_ave = 0.0
109     energy2_ave = 0.0
110     magnetization_ave = 0.0
111     magnetization2_ave = 0.0
112
113     ! Set up the initial spin configuration.
114     select case(ConfigType)
115         case(1) ! checkerboard setup
116             A(1,1) = 1
117             do i = 1, nrows+1
118                 A(i+1,1) = -A(i,1)
119             enddo
120             do j = 1, ncols+1
121                 A(:,j+1) = -A(:,j)
122             enddo
123             ! (note: the requirement that nrows and ncols are even is to
124             ! ensure that the first two rows/cols start out the same as the
125             ! last two)
126         case(2) ! interface
127             do i = 1, nrows+2
128                 do j = 1, (ncols+2)/2
129                     A(i,j) = 1
130                 enddo
131                 do j = (ncols+2)/2 + 1, ncols+2
132                     A(i,j) = -1
133                 enddo
134             enddo
135         case(3) ! unequal interface
136             do i = 1, nrows+2
137                 do j = 1, (ncols+2)/4
138                     A(i,j) = 1
139                 enddo
140                 do j = (ncols+2)/4 + 1, ncols+2
141                     A(i,j) = -1
142                 enddo
143             enddo
144         case default
145             print*, "Error! Check ConfigType parameter in ising.in"
146             stop
147     end select
148
149     ! Main loop containing Monte Carlo algorithm:
150     MC_passes: do ipass = 0, npass
151
152         ! If MovieOn is .true., write the spin array to an output every
153         ! npass/50 steps.
154         if ((MovieOn) .and. (mod(ipass,npass/50) == 0)) then

```

```

155     do i = 2, nrows+1
156         do j = 2, ncols+1
157             write(32,*) A(i,j)
158         enddo
159     enddo
160 endif
161
162 ! If ipass is greater than nequil (the number of equilibration steps),
163 ! calculate the magnetization and energy:
164 if (ipass > nequil) then
165     output_count = output_count + 1
166     magnetization = sum(A(2:nrows+1,2:nrows+1))/(ncols*nrows*1.0)
167     magnetization_ave = magnetization_ave + magnetization
168     magnetization2_ave = magnetization2_ave + magnetization**2
169     energy = 0.0
170     do i = 2, nrows + 1
171         do j = 2, ncols + 1
172             energy = energy - A(m,n)*(A(m-1,n)+A(m+1,n)+A(m,n-1)+A(m,n+1))
173         enddo
174     enddo
175     ! Divide the energy by the total number of spins to get the ave
176     ! energy per spin, and divide by 2 to account for double counting.
177     energy = energy/(ncols*nrows*2.0)
178     energy_ave = energy_ave + energy
179     energy2_ave = energy2_ave + energy**2
180 endif
181
182 ! Randomly choose a spin to change:
183 m = nint((nrows-1)*ran1(5) + 2) ! choose a random row
184 n = nint((ncols-1)*ran1(5) + 2) ! choose a random column
185 trial_spin = -A(m,n) ! trial spin value
186
187 ! Find change in energy (deltaU) due to trial move.
188 ! If exp(-beta*deltaU) > eta, where eta is random, accept move:
189 deltaU = -trial_spin*(A(m-1,n)+A(m+1,n)+A(m,n-1)+A(m,n+1))*2
190 log_eta = dlog(ran1(5) + 1.0d-10) ! random number 0-1 (+ tiny offset)
191 if (-beta*deltaU > log_eta) then
192     A(m,n) = trial_spin
193     if (m == 2) A(nrows+2,n) = trial_spin
194     if (m == nrows+1) A(1,n) = trial_spin
195     if (n == 2) A(m,ncols+2) = trial_spin
196     if (n == ncols+1) A(m,1) = trial_spin
197 endif
198
199 enddo MC_passes
200
201 ! Write final spin array to output file
202 if (.not. MovieOn) then
203     do i = 2, nrows + 1
204         do j = 2, ncols + 1
205             write(32,*) A(i,j)
206         enddo
207     enddo

```



```

208     endif
209     write(33,*) temp, abs(magnetization_ave/output_count), &
210         magnetization2_ave/output_count, &
211         beta*(magnetization2_ave/output_count - (magnetization_ave/output_count)**2)
212     write(34,*) temp, energy_ave/output_count, energy2_ave/output_count, &
213         (beta**2)*(energy2_ave/output_count - (energy_ave/output_count)**2)
214
215     enddo scan_loop
216
217     close(32)
218     close(33)
219     close(34)
220
221     print*, "Program ising.f90 complete!"
222     print*, "Look at 'spin-array' with IDL program see_spins.pro"
223
224     contains
225
226
227     !_____RANDOM NUMBER GENERATING FUNCTION_____!
228
229     double precision function ran1(idum)
230     implicit none
231     double precision :: r(97)
232     integer, intent(IN) :: idum
233     save
234     integer, parameter :: M1=259200, IA1=7141, IC1=54773
235     real, parameter :: RM1=1.0d0/M1
236     integer, parameter :: M2=134456, IA2=8121, IC2=28411
237     real, parameter :: RM2=1.0d0/M2
238     integer, parameter :: M3=243000, IA3=4561, IC3=51349
239     integer :: IX1, IX2, IX3, jjj
240     integer :: iff=0
241     if (idum < 0 .or. iff == 0) then
242         iff = 1
243         IX1 = mod(IC1-idum,M1)
244         IX1 = mod(IA1*IX1+IC1,M1)
245         IX2 = mod(IX1,M2)
246         IX1 = mod(IA1*IX1+IC1,M1)
247         IX3 = mod(IX1,M3)
248         do jjj = 1,97
249             IX1 = mod(IA1*IX1+IC1,M1)
250             IX2 = mod(IA2*IX2+IC2,M2)
251             r(jjj) = (dfloat(IX1)+dfloat(IX2)*RM2)*RM1
252         end do
253     end if
254     IX1 = mod(IA1*IX1+IC1,M1)
255     IX2 = mod(IA2*IX2+IC2,M2)
256     IX3 = mod(IA3*IX3+IC3,M3)
257     jjj = 1+(97*IX3)/M3
258     if (jjj > 97 .or. jjj < 1) PAUSE
259     ran1 = r(jjj)
260     r(jjj) = (dfloat(IX1)+dfloat(IX2)*RM2)*RM1

```

```

261 end function ran1
262
263 end program ising

```

This is the required input file for the above program:

```

1 nrows - number of rows of spins (even number)
2 20
3 ncols - number of columns of spins (even number)
4 20
5 npass - number of passes for each temperature
6 200000
7 nequil - number of equilibration steps for each temperature
8 100000
9 high_temp - temperature to start scan at
10 2.92
11 low_temp - temperature to finish scan at
12 0.92
13 temp_interval - scanning interval
14 .1
15 ConfigType - 1: checkerboard, 2: interface, 3: unequal interface
16 1
17 MovieOn - set to .true. when running for 1 temp to make movie
18 .false.
19 End of file.

```

This is the IDL helper program for visualizing the final spin arrays at each temperature:

```

1 pro see_spins
2
3 inputfile = 'spin-array'
4 openr, inlun, inputfile, /get_lun
5 readf, inlun, nrows
6 readf, inlun, ncols
7 readf, inlun, nframes
8 readf, inlun, MovieOn
9 print, "MovieOn is", MovieOn
10 A = intarr(ncols,nrows)
11 window, 5, xsize=ncols*20, ysize=nrows*20, $
12   title='2D Ising Model: light = +, dark = -'
13 for n = 0, nframes-1 do begin
14   for i = 0, nrows-1 do begin
15     for j = 0, ncols-1 do begin
16       readf, inlun, s
17       A(j,nrows-1-i) = s
18     endfor
19   endfor
20   if (MovieOn eq 2) then begin
21     if (total(A) < 0) then A = -A
22     for i = 0, nrows-1 do begin
23       for j = 0, ncols-1 do begin
24         if (A(j,nrows-1-i) eq -1) then A(j,nrows-1-i) = 1 $
25           else A(j,nrows-1-i) = -1

```

```

26     endfor
27     endfor
28 endif
29 A = A*1000
30 A = congrid(A, ncols*20, nrows*20)
31 tv, A
32 A = intarr(ncols, nrows)
33 print, "Frame", n
34 wait, 0.1
35 endfor
36 free_lun, inlun
37
38 end

```

Onsager's Exact Solution

I happened to find this while I was looking for information for my presentation, and I thought it was somewhat amusing.

In 1942, Onsager developed an exact solution to the problem of Ising spins in a plane, the “two-dimensional Ising model.” This work stands, to this day, as a pinnacle of the achievements of theoretical physics of our time. Onsager’s solution yielded the thermodynamic properties of the interacting system, and demonstrated the phase transition at T_c but in a form quite unlike that of Curie-Weiss. In particular, the infinite specific-heat anomaly at T_c is a challenge for approximate, simpler theories to reproduce. Onsager’s discovery was not without an amusing sequel. The original solution was given by Onsager as a discussion remark, following a paper presented to the New York Academy of Science in 1942 by Gregory Wannier, but the paper, based on an application of Lie algebras, only appeared two years later. However, his formula for the spontaneous magnetization below T_c which requires substantial additional analysis, $M = (1 - x^{-2})^{1/8}$, $x = \sinh(2J_1/kT) \sinh(2J_2/kT)$, was never published by him, but merely “disclosed.” It required four years for its decipherment. It was first exposed to the public on 23 August 1948 on a blackboard at Cornell University on the occasion of a conference on phase transitions. Laslo Tisza had just presented a paper on The General Theory of Phase Transitions. Gregory Wannier opened the discussion with a question concerning the compatibility of the theory with some properties of the Ising model. Onsager continued this discussion and then remarked that – incidentally, the formula for the spontaneous magnetization of the two-dimensional model is just that (given above.) To tease a wider audience, the formula was again exhibited during the discussion which followed a paper by Rushbrooke at the first postwar IUPAP statistical mechanics meeting in Florence in 1948; it finally appeared in print as a discussion remark. However, Onsager never published his derivation. The puzzle was finally solved by C.N. Yang and its solution published in 1952. Yang’s analysis is very complicated ...

— D.C. Mattis, in *The Theory of Magnetism I*